
Introduction

The purpose of this tutorial is to help you learn the basics of Excel Visual Basic for Applications (VBA).  This tutorial addresses the topics of Macro Recording, the Visual Basic Editor (VBE), variable data types, and how to begin a program.
Note:  If you have not read the “General Notes” document, then reading the following (and subsequent) tutorials may leave you without the common conventions and general information regarding the tutorials.  Please read “General Notes” before continuing on with this tutorial.

Terminology

· Macro – a user-defined command that is part of an application (in this case, Excel) and executes a series of instructions
· Syntax – the structure of a defined language (Excel VBA)
· Objects – separate parts with specific operations available for repeated use.  (Objects are similar to the subject in a sentence; objects can be described (properties) and they can have actions done to them (methods))
· Properties – descriptions that define unique aspects of an object.  (Properties are similar to adjectives in a sentence)
· Methods – actions performed on an object.  (Methods are similar to the verbs in a sentence)
· Action – a combination of objects, properties, and methods that comprise an interpretable syntax
· VBE – Visual Basic Editor (XLS: Developer | Code | Visual Basic) is the interface where syntax is typed and edited for your program
· Variable – any factor that you want to change in the program
Best Friends
There are some notable references that are available to you for programming with VBA
· VBE Error Messages

· Excel Macro Recorder (XLS: Developer | Code | Record Macro)

· VBE Help (VBE: F1) and Excel Help (XLS: F1)
· Object Browser (VBE: View | Object Browser (F2))
· Internet Search Engines and Microsoft Software Developer Network (MSDN)
These tools are helpful for learning the syntax of VBA.  VBA is proprietary in how the code is written, and the editor will let you know when incorrect syntax has been coded.  
The editor uses error and program end/debug messages to let you know when you have made a mistake.  The end/debug message box allows you to end the program or debug the program in the event errors occur.  If you debug the program, then VBE will open and highlight (in yellow) the line of syntax in which the error occurred.  When you debug the program, you can change syntax, continue the program, or stop the program, among other things.
The Excel Macro Recorder records almost every action you perform within the Excel application itself.  For example, the recorder will record save, open, and close operations; row and column insertion and deletion; and formula building.  The recorder will provide a good base to determine how to perform certain actions without having to search the help documentation or the web for syntax.
VBE and Excel Help also provides examples of syntax for specific actions.  Help might seem fickle and hard to use because the search tool is by no means as encompassing as an internet search engine, but it is very helpful and too often overlooked as source documentation.  Your best hint with the Help search is to try multiple synonyms for your search word(s) in the event that your first search doesn’t return relevant help files.
The Object Browser is also a good place to consult what objects, properties, methods, events, and constants are available for use.  When you open the browser you will see "Classes" and "Members of 'Classes'".  Classes are objects.  For example, if you select the Range Class (or object), you will notice that Members includes properties, such as Address, Column Width, and Horizontal Alignment, and methods, such as Clear, Copy, and Cut.  The Object Browser functionality also appears in the Intellisense, a built-in mechanism in VBE that automatically shows you available properties, methods, events, and contants for your objects.  Intellisense is a cool tool!  (For example, if you were to look at Sub FirstTime below, and let’s image that you typed Range("B4"), then as soon as you type the “.” (or “dot”), the Intellisense would appear.  You can scroll through Intellisense, type partial words to jump to words within Intellisense, and auto complete words by pressing <Tab> or <Enter>).
Internet search engines are useful to query key word(s) for syntax solutions.  Include something like “VBA” or “Excel VBA” with your key word(s) to better define the internet search string; otherwise you might end up getting some Java, C++, C#, or other programming syntax.  MSDN has an extensive library and a number of community forums that people post to regularly.  If you have a question, I’m sure an MSDN forum will have an answer because it’s highly likely that someone has already posted the same question and others have answered that question.  Use all of the aforementioned tools to your advantage in your quest to learn VBA.
Baby Steps
Let’s run through a simple example of how to record a macro.  Start the Macro Recorder (XLS: Developer | Code | Record Macro).  Give your macro a name and click OK.  (See the “Record Macro Dialog Box” section in the Appendix for more detail regarding the macro dialog box options).  VBE is now recording the actions you perform within Excel.  Click cell B4, type your first name, and hit enter.  In cell B5, type your last name and hit Enter.  Click cell B5 and insert a row (XLS: Home | Cells | Insert | Insert Sheet Rows).  Click cell B4.  Finally, stop the Macro Recorder (XLS: Developer | Code | Stop Macro Recording).
In order to access the recorded code, go to XLS: Developer | Code | Macros.  A dialog box should appear with the name of your macro in it.  Click your macro and click the Edit button.  This will automatically open VBE and show your code.  Your code should look something like the following:

Sub FirstTime()

Range("B4").Select

    ActiveCell.FormulaR1C1 = "Matthew"

Range("B5").Select

    ActiveCell.FormulaR1C1 = "Herbert"

Range("B5").Select

    Selection.EntireRow.Insert

Range("B4").Select

End Sub

This is pretty good for a first-time macro, but what does all of that code mean?  VBA is an object oriented language, so as mentioned earlier, you take objects and do things to them.  Range("B4") is an object and Select is a method performed on Range("B4").  ActiveCell is another object.  FormulaR1C1 is a property of ActiveCell and the property is set to the text Matthew.  Lastly, Selection is an object, EntireRow is a property of Selection, and Insert is a method performed on the EntireRow of the Selection (i.e. take the selection and insert rows).
There are at least two areas wherein the above syntax could be better – .Select and .FormulaR1C1.  The .Select and .Activate methods are time consuming methods.  The only time I use these two methods are when I want to set the spreadsheet focus to a certain location prior to the macro finishing (i.e. it’s the last line of code prior to End Sub).  One way to condense the first two lines of the macro, and to avoid .Select, is to replace the text with Range("B4").Formula = "Matthew".  .FormulaR1C1 is one of the two possible formula notations within Excel; A1 notation (.Formula) is the other.  I don’t like R1C1 notation; I prefer A1 notation.  Though the macro recorder records in R1C1 notation, you don’t have to use it.  ActiveCell.FormulaR1C1 = "Matthew" could equally read as ActiveCell.Formula = "Matthew" or even ActiveCell.Value = "Matthew".  Confused?  Don’t be.  This will sort itself out.
As a side note, I will point out at least three very important debugging options within VBE:  (1) Step Into (VBE: Debug | Step Into (F8)), (2) Debug.Print, and (3) Breakpoints (VBE: Debug | Toggle Breakpoint (F9)).  
1. Step Into will allow you to evaluate your code one line at a time.  Each time you press F8, a yellow “breakpoint” will advance one line of code.  As you press F8, you can assess how the program behaves.
2. Debug.Print will allow you to print to the Immediate Window (VBE: View | Immediate Window).  So, if you created a variable called strFirstName (i.e. Dim strFirstName As String) and assigned the variable as Matthew (i.e. strFirstName = "Matthew"), then you could use the following syntax to print the value to the Immediate Window:  

Debug.Print strFirstName
As you run the macro, the Debug.Print line will print “Matthew” in the Immediate Window.  This is a useful way to see how your variables are behaving in the program.  (One alternative is to mouse over the variable and view the small text box that shows the variable’s value; however, mouse over can get old fast, so Debug.Print is a bit easier to use).
3. Breakpoints will allow you to “break,” or pause, program execution on the designated line(s) of code.  This is useful because it can be used to quickly run through sections of code that are already tested and still allow you to stop the program at a point where you need to continue evaluating the code.

These debugging tools will come in handy as you start to develop your programs, so take note of these three basic tools.

Go back to Excel, delete your first and last name, and run your Macro (XLS: Developer | Code | Macros | <Your Macro> | Run).  Simple, right?  
Baby Steps Part II

Let’s run through another simple example of how to record a macro.  At the end of the tutorial you’ll find a section entitled “Supplemental Information.”   This is where you will find the data necessary for this example.  Either access this supplemental information now (via the Excel workbook) or type this information into a blank worksheet.
Your little brother’s friend, Tim, is trying to understand how interest and time affect the amount of money he could borrow for a car.  He has set up a simple spreadsheet that utilizes the payment function =PMT(rate, nper, pv, [fv], [type]).  (The parameters enclosed in brackets are optional, i.e. [fv] and [type] are optional in the PMT function).  You check the formulas in rows 12, 15, and 17 to make sure the functions are correct.  Cell C12 reads =C11/C9, cell B15 reads =C9*C14, and cell B17 reads =PMT(C12, C15, -C10).  The functions are correct for the rest of the columns too.  (If you need help with the predefined Excel functions, such as PMT, search the Excel Help for further information and sample examples).
In viewing the numbers closer, you wonder why Tim has not entered a loan amount for the four separate scenarios. You ask Tim why he has left the amounts blank and he replies, “Because I can afford a $600 payment and I want the payment to always be $600 for each scenario.”  Tim then asks you if you know how to “back” into the loan amount given the fixed interest rate and number of periods.  Currently, Tim is using a guess and check method by changing the loan amount (i.e. entering values in row 10) until he sees the payment amount reach $600.  Luckily, you are familiar with the goal seek option.  You share this knowledge with Tim by showing him how to perform goal seek for cell C10, so that Tim doesn’t have to perform the guess and check method.
You perform the following steps: XLS: Data | Data Tools | What-If Analysis | Goal Seek; Set Cell: C17, To Value: 600, By Changing Cell: C10; Click OK.  Tim is amazed at the power of Excel and how quickly Excel solves the answer.  You simply note that all he needs to do now is repeat the process for each column.  Tim follows your directions.  After observing the loan amounts, Tim changes the interest rate (row 11) and number of years (row 14).  As a result, the payment for each scenario changes as well (provided that the calculation mode is set to automatic); however, Tim wants the payment to remain at a constant $600.  He knows that he can perform the goal seek option again, but he’ll have to do it four more times.  Tim comes back to you and asks if there is an easy way to repeat the process over and over again because he plans on creating multiple “what-if” scenarios.  Luckily, you have one possible answer.
Follow the next three steps:

1. Start the macro recorder (XLS: Developer | Code | Record Macro) and give your macro a name, e.g. SeekLoanAmount.  (Note: you can’t have spaces or other non-savable characters, such as ? , / < > in the name).
2. Run Goal Seek on column C (XLS: Data | Data Tools | What-If Analysis | Goal Seek; Set Cell: C17, To Value: 600, By Changing Cell: C10, Click OK).  You will run Goal Seek a total of 4 times, once for each column.
3. Stop the Macro Recorder (XLS: Developer | Code | Stop Recording).

Your recorded macro should look something like the following:

Sub SeekLoanAmount()

Range("C17").GoalSeek Goal:=600, ChangingCell:=Range("C10")
Range("D17").GoalSeek Goal:=600, ChangingCell:=Range("D10")
Range("E17").GoalSeek Goal:=600, ChangingCell:=Range("E10")
Range("F17").GoalSeek Goal:=600, ChangingCell:=Range("F10")

End Sub

Now Tim can change the interest rate and number of years as many times as he would like and run the goal seek option by simply running the macro (or by performing the tedious task of running goal seek four times).  You instruct Tim to run his macro by going to XLS: Developer | Macros | <Your Macro> | Run.  Go ahead and change the interest rate and number of years, observe what happens, and run your macro.  Tim thanks you for helping him save so much time on his analysis.
Variables:  Data Types, a Quick Introduction to a More Robust Way of Programming
Variables are used to change target factors.  A characteristic of a variable is a data type; the data type determines what kind of data the variable can store and how much memory the computer will allocate for the given variable.  Data types include Byte, Boolean, Integer, Long, Currency, Decimal, Single, Double, Date, String, Object, Variant (default), user-defined types, and object types.  There are many data types, more than are listed here!
Commonly used data types are (search “Data Type Summary” in the VBE Help for a complete explanation):
· Integer/Long – a non-decimal number, e.g. 1; 24; 32,456; integer ranges from -32,768 to 32,768; long ranges from -2,147,483,648 to 2,147,483,648
· Single/Double – a large non-integer number, e.g. 1.233, 4.5569000; single ranges from 
-3.402823e38 to -1.401298e-45 for negative numbers and 1.401298e-45 to 3.402823e38 for positive numbers.  (Double holds MANY more place holders!)
· String – variables that contain text.  Variable length strings can hold up to 2 billion characters and fixed length strings can hold up to 65,400 characters
· Variant – a special data type that can contain any kind of data EXCEPT fixed-length string data.  Numeric data can be any integer or real number.  Use this data type if you do NOT know exactly how to declare your variable or if you need the variable to handle multiple data types.  (For example, maybe you have varMatch = Application.Match(intVal, intArr, 0) wherein the Match function will return a result or an error – two completely different data types)
Beginning a Program and Dimensioning Variables
If you are writing a program that requires variables, then you must declare your variables in order to make them available for use in your program.  However, before declaring your variable you must begin a program.  You can do this by opening VBE (XLS: Developer | Code | Visual Basic).  You will notice the Project Explorer on the left-hand sidebar.  The Project Explorer has a tree view structure for your files.  The bolded names are the names of workbooks and/or add-ins.  When you expand the current Excel file (i.e. click the “+” sign next to the name of the file you are working with) you will notice a “Microsoft Excel Objects” and “Modules” folder.  The “Modules” folder will not be present if you have not recorded a macro or inset a Module.  As long as the focus within the Project Explorer is on the file you are working with (or any subfolder/file within the parent file), you can insert a module by doing the following:  VBE: Insert | Module.  Within Microsoft Excel Objects, you will have worksheets and “ThisWorkbook”.  Within Modules, you will have modules.  Modules are named by default as Module1, Module2, etc.  You can rename your model in the Properties Window.  Go ahead and insert a module.  Then click inside the Code Window (VBE: View | Code).  You can create either Sub procedures or Functions inside the Code Window.
Sub ProcName()


'Code
End Sub

Function FunctionName()


'Code
End Function

The name of your procedure or function will be in place of the ProcName/FunctionName above.  The program itself will be between the Sub/Function and End Sub/End Function lines (i.e. where the comment Code is).  Comments are made by placing an apostrophe before your comment.  Commented text is NOT evaluated by the program.  I recommend that you start a good habit of using comments within your code as well as indenting properly.  Good comments and proper indentation will save you future debugging headaches.  See “Illustrative Code – Comments and Indentation” in the Appendix for an illustrative example of comments and indentation.
The following macro doesn’t do anything more than dimension variables and include comments:

Sub DimVariables()

Dim myInt As Integer    'integer data type

Dim myLng As Long       'long data type

Dim mySng As Single     'single data type

Dim myDbl As Double     'double data type

Dim myStr As String     'string data type

Dim myVar As Variant    'variant data type

Dim Var                 'variant data type

End Sub
This program ONLY consists of variables and nothing else.  The Dim stands for dimension.  Dim is a key word that signals to the program that the text following Dim is associated with a variable.  The As Type signals to the program what data type the variable is.  At this point the program is ready to use your variables (myInt, myLng, mySng, myDbl, myStr, myVar, Var).  
*Note: If you can think it, you can write a program to do it.  Let the computer work for you.

APPENDIX
Record Macro Dialog Box
The Record Macro Dialog Box has the following options:

· Macro name:  The actual name of your macro, which by default will be Macro1 for the first time that you record a macro, Macro2 for the second macro, etc.  Choose a name that will be descriptive enough for you to remember, e.g. InsertFirstLastName
· Shortcut key:  Keyboard combinations that allow you to execute the macro via key strokes rather than XLS: Developer | Code | Macros.  The dialog will allow Ctrl+ and Ctrl+Shift+ as keyboard shortcut prefixes; VBA adds Alt+ to the mix
· Store macro in:
· This Workbook.  The macro will be placed in a standard module within your current workbook

· New Workbook.  A new workbook will be created and the macro will be placed in a standard module within the new workbook
· Personal Macro Workbook.  The PERSONAL workbook is a file located within the XLSTART folder.  This PERSONAL file will not exist if you have never recorded a macro in the Personal Macro Workbook.  Each time Excel starts, it opens all the files located in the XLSTART folder.  The PERSONAL file is hidden by default, so though it is open, the user is unable to view it unless the user unhides the file (XLS: View | Window | Unhide).  Since this file opens each time you start a session of Excel, you can place commonly used procedures and functions in the PERSONAL file
· Description:  Descriptive, commented text that is placed at the beginning of the macro
Supplemental Information

xBuy A Car x

Scenario

Description One Two Three Four

8

9

Period 12 12 12 12

10

Loan Amount – – – –

11

Interest Rate 5.0% 4.0% 3.0% 2.0%

12

Period Interest 0.4% 0.3% 0.3% 0.2%

13

14

Number of Years 5 5 5 5

15

Number of Periods 60 60 60 60

16

17

Payment – – – –

19

21

xGoal Seek Settings x

22

23

XLS: Data | Data Tools | What-If Analysis | Goal Seek

24

25

Set Cell: C17 D17 E17 F17

26

To Value: 600 600 600 600

27

By Changing Cell: C10 D10 E10 F10

28

29

xEND x


Illustrative Code – Comments and Indentation
You can insert the code below into a module for easier viewing.  Sub TestFirstAlphaFunction can be run and the results can be viewed in VBE’s Immediate Window.  The comments for the functions should be clear enough, so go ahead and read the comments below and take note of the indentations.  The code below is one way of commenting code.  Again, pick a style and be consistent.

FirstAlpha has three parameters – strText (required), blnReverse (optional), and lngStart (optional) – and returns a Long data type as its return value.  The underscore is a line continuation character, so the three lines of syntax – starting with Private Function and ending with Long – are viewed by the compiler as one complete line of syntax.  FirstAlphaLoop is set up in a similar manner to FirstAlpha.
Sub TestFirstAlphaFunction()

Dim strText As String

Dim lngVal As Long

strText = "109B54D"

lngVal = FirstAlpha(strText)

    Debug.Print "Forward Direction - lngVal:"; lngVal

lngVal = FirstAlpha(strText, True)

    Debug.Print "Reverse Direction - lngVal:"; lngVal

End Sub

Private Function FirstAlpha(strText As String, _

                            Optional blnReverse As Boolean = False, _

                            Optional lngStart As Long) As Long

'---------------------------------------------------------------------

'INFO:      07/01/2009, Matthew Herbert

'---------------------------------------------------------------------

'PURPOSE:   This function will find the first alpha character within

'           the specified string.  Returns zero if no alpha character

'           is found.

'

'           SINCE THE FUNCTION IS PRIVATE, NO MEASURES ARE TAKEN FOR

'           lngStart < 0 OR lngStart > Len(strText) BECAUSE THE AUTHOR

'           KNOWS THAT THESE ARE INVALID ENTRIES!  (I may modify this

'           at a later date).

'

'strText    The string that is searched for the alpha character.

'

'blnReverse This parameter is optional and is set to FALSE.  TRUE will

'           search from right to left and FALSE will search from

'           left to right.

'

'lngStart   If so desired, the starting position of the search.

'---------------------------------------------------------------------

'UTILIZES:  FirstAlphaLoop Function.

'---------------------------------------------------------------------

'NOTES:     See FirstAlphaLoop for the Chr/Char character codes.

'---------------------------------------------------------------------

'Reverse direction

If blnReverse Then

    'get the starting position

    If lngStart = 0 Then lngStart = Len(strText)

    'return the alpha character position

    '   lngStart > lngEnd

    FirstAlpha = FirstAlphaLoop(strText, lngStart, 1, -1)

'Forward direction

Else

    'get the starting position

    If lngStart = 0 Then lngStart = 1

    'return the alpha character position

    '   lngStart < lngEnd

    FirstAlpha = FirstAlphaLoop(strText, lngStart, Len(strText), 1)

End If

End Function

Private Function FirstAlphaLoop(strText As String, _

                                lngStart As Long, _

                                lngEnd As Long, _

                                intStep As Integer) As Long

'---------------------------------------------------------------------

'INFO:      07/01/2009, Matthew Herbert

'---------------------------------------------------------------------

'PURPOSE:   Return the first alpha character based on the For...Next

'           loop structure.  Note the following, which is based on the

'           FirstAlpha function:

'

'           blnReverse = True Then

'               Evaluate the loop right to left (i.e. Step -1)

'

'           blnReverse = False Then

'               Evaluate the loop left to right (i.e. Step 1)

'

'strText    The string that is searched for the alpha character.

'

'lngStart   The starting position for the For...Next loop.

'

'lngEnd     The ending position for the For...Next loop.

'

'intStep    The step for the For...Next loop.

'---------------------------------------------------------------------

'NOTES:     Remember that if intStep is < 0 then lngStart needs to
'           be > lngEnd in the the For...Next loop.

'

'Chr/Char:  Returns a String containing the character associated

'VBE/XLS    with the specified character code.

'

'Asc:       Returns an Integer representing the character code

'           corresponding to the first letter in a string.

'

'Chr Range: 1 To 255

'

'Alpha:     65 To 90  = UPPER case letters

'           97 To 122 = lower case letters

'

'Numbers:   48 To 57  = 0 To 9, respectively

'---------------------------------------------------------------------

Dim Lng As Long

'loop through the characters

For Lng = lngStart To lngEnd Step intStep

    'get the integer that corresponds to the character

    Select Case Asc(Mid(strText, Lng, 1))

        'determine if the integer is an alpha integer

        Case 65 To 90, 97 To 122

            'return the value to the function

            FirstAlphaLoop = Lng

            'exit the loop

            Exit For

    End Select

Next Lng

End Function

